Abstract Governing Equations for Dark Energy–Dark Matter Interactions

Using xPert

In this notebook we derive all equations to 0th and 1st order of dark energy in the form of a scalar field ϕ that interacts with dark matter, where the dark matter mass $m = m(\phi)$.

Load xPert package

Mathematica 14.3 introduces commands that are already assigned in xAct . The following lines allow xAct to use its own definition .

```
In[1]:= Unprotect[Commutator, Anticommutator];
ClearAll[Commutator, Anticommutator];
Remove[Commutator, Anticommutator];
```

```
In[4]:= << xAct`xPert`
```

```
------------------------------------------------------------
  --
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (c) 2003-2020, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
  --
Package xAct`xTensor`  version 1.2.0, {2021, 10, 17}
CopyRight (c) 2002-2021, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
  --
Package xAct`xPert`  version 1.0.6, {2018, 2, 28}
CopyRight (c) 2005-2020, David Brizuela, Jose M.
  Martin-Garcia and Guillermo A. Mena Marugan, under the General Public License.
------------------------------------------------------------
  --
These packages come with ABSOLUTELY NO WARRANTY; for details
  type Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
  --
** Variable $PrePrint assigned value ScreenDollarIndices
** Variable $CovDFormat changed from Prefix to Postfix
** Option AllowUpperDerivatives of ContractMetric changed from False to True
** Option MetricOn of MakeRule changed from None to All
** Option ContractMetrics of MakeRule changed from False to True
```

```
In[5]:= $DefInfoQ = False;
```

In[6]:=
```
$PrePrint = ScreenDollarIndices;
```

Derive abstract equations

We define a 4D manifold *M*

In[7]:=
```
DefManifold[M, 4, IndexRange[a, n]]
```

and an abstract metric *g*

In[8]:=
```
DefMetric[-1, background[-a, -b], cd, PrintAs -> "g"]
```

with the perturbations of the metric stored as *h*

In[9]:=
```
DefMetricPerturbation[background, pert, ϵ, PrintAs -> "h"]
```

Lagrangian

In our model of DM–DE interactions, the DM mass m depends on the scalar field ϕ. The Lagrangian of our model that otherwise corresponds to ΛCDM contains the terms $L \supset -\frac{\nabla^\mu \phi \nabla_\mu \phi}{2} - V(\phi) - \rho_{\rm dm}$. In the following, we set up our system to make computations using this Lagrangian, including perturbations.

In[10]:=
```
DefTensor[sf[], M, PrintAs -> "ϕ"] (* our scalar field *)
DefTensorPerturbation[pertsf[LI[order]], sf[], M, PrintAs -> "δϕ"]
(*The scalar field perturbations*)
```

In[12]:=
```
DefConstantSymbol[massP, PrintAs -> "m_p"](*The Planck mass*)
DefScalarFunction[mf, PrintAs -> "m"](*The DM mass,
defined as a scalar function to realise the ϕ dependence*)
DefScalarFunction[V](*The scalar field potential*)
```

In[15]:=
```
DefTensor[nf[], M, PrintAs -> "n"](*The DM number density*)
DefTensorPerturbation[pertnf[LI[order]], nf[], M, PrintAs -> "δn"]
(*The DM number density perturbation*)
```

Now, we have all ingredients to define our Lagrangian as $L = \sqrt{-g}\left(-\frac{\nabla^\mu \phi \nabla_\mu \phi}{2} - V(\phi) - \rho_{\rm dm}\right)$:

In[17]:=
```
L = Sqrt[-Detbackground[]] (-1/2 cd[-b][sf[]] × cd[b][sf[]] - V[sf[]] - nf[] × mf[sf[]])
```

Out[17]=

$$\sqrt{-\tilde{\tilde{g}}}\left(-m[\phi]\, n - V[\phi] - \frac{1}{2}(\nabla_b \phi)(\nabla^b \phi)\right)$$

The variation of the Lagrangian is given by:

In[18]:=
```
varL = L // Perturbation // ExpandPerturbation // ContractMetric // ToCanonical
```

Out[18]=

$$-\frac{1}{2}\sqrt{-\tilde{\tilde{g}}}\; m[\phi]\, n\, h^{1b}{}_{b} - \sqrt{-\tilde{\tilde{g}}}\; m[\phi]\, \delta n^{1} - \frac{1}{2}\sqrt{-\tilde{\tilde{g}}}\; h^{1b}{}_{b}\, V[\phi] - \sqrt{-\tilde{\tilde{g}}}\,(\nabla_b \phi)\, \delta\phi^{1;b} + \frac{1}{2}\sqrt{-\tilde{\tilde{g}}}\; h^{1}{}_{ba}(\nabla^a \phi)(\nabla^b \phi) - \frac{1}{4}\sqrt{-\tilde{\tilde{g}}}\; h^{1a}{}_{a}(\nabla_b \phi)(\nabla^b \phi) - \sqrt{-\tilde{\tilde{g}}}\; n\, \delta\phi^{1}\, m'[\phi] - \sqrt{-\tilde{\tilde{g}}}\; \delta\phi^{1}\, V'[\phi]$$

Equation of Motion

The equations of motion are derived from the Lagrangian as $0 = \frac{\delta L}{\delta\phi(t,x,y,z)} - \partial_\mu \frac{\delta L}{\delta\left(\partial_\mu \phi(t,x,y,z)\right)}$.

Background

The background equation of motion is the modified Klein–Gordon equation $0 = \nabla_\mu \nabla^\mu \phi - V' - \rho'_{\rm dm}$, which can either be obtained by taking the derivatives with respect to the background, such that $0 = \frac{\delta L^{(0)}}{\delta\bar\phi(t)} - \partial_t \frac{\delta L^{(0)}}{\delta\left(\partial_t \bar\phi(t)\right)}$, which yields

In[19]:=
```
VarD[sf[], cd][L] / Sqrt[-Detbackground[]] // Simplify
```

Out[19]=

$$\nabla_a \nabla^a \phi - \mathrm{nm}'[\phi] - \mathrm{v}'[\phi]$$

or by taking the derivatives with respect to the first order perturbations, such that

$0 = \frac{\delta L^{(1)}}{\delta(\delta\phi(t,x,y,z))} - \partial_\mu \frac{\delta L^{(1)}}{\delta\left(\partial_\mu \delta\phi(t,x,y,z)\right)}$, which yields

In[20]:=
```
eom0 =
 VarD[pertsf[LI[1]], cd][varL] / Sqrt[-Detbackground[]] /. delta[-LI[1], LI[1]] → 1 // Simplification
```

Out[20]=

$$\nabla_a \nabla^a \phi - \mathrm{nm}'[\phi] - \mathrm{v}'[\phi]$$

First-Order Perturbations

The latter approach can be used for any higher order of perturbations, e.g. for the first order perturbations we take the second order Lagrangian

In[21]:=
```
varL2 = ToCanonical[ContractMetric[ExpandPerturbation[Perturbation[L, 2]]]] /
   Sqrt[-Detbackground[]] // Simplification
```

Out[21]=

$$\frac{1}{8}\Big(2\,\mathrm{m}[\phi]\Big(2\,\mathrm{n}\,h^{1}{}_{ba}\,h^{1ba} - 2\,\mathrm{n}\,h^{2b}{}_{b} - 2\,h^{1b}{}_{b}\left(\mathrm{n}\,h^{1a}{}_{a} + 2\,\delta\mathrm{n}^{1}\right) - 4\,\delta\mathrm{n}^{2} + \mathrm{n}\left(h^{1a}{}_{a}\right)^2\Big) - 4\,h^{1a}{}_{a}\,h^{1b}{}_{b}\,\mathrm{v}[\phi] -$$
$$4\,h^{2b}{}_{b}\,\mathrm{v}[\phi] + 2\left(h^{1a}{}_{a}\right)^2\mathrm{v}[\phi] - 8\,\delta\phi^{1}{}_{;b}\,\delta\phi^{1;b} - 8\,h^{1a}{}_{a}\left(\nabla_b\phi\right)\delta\phi^{1;b} - 8\left(\nabla_b\phi\right)\delta\phi^{2;b} -$$
$$8\,h^{1}{}_{ac}\,h^{1}{}_{b}{}^{c}\left(\nabla^a\phi\right)\left(\nabla^b\phi\right) + 4\,h^{2}{}_{ba}\left(\nabla^a\phi\right)\left(\nabla^b\phi\right) + 2\,h^{1}{}_{ac}\,h^{1ac}\left(\nabla_b\phi\right)\left(\nabla^b\phi\right) -$$
$$2\,h^{1a}{}_{a}\,h^{1c}{}_{c}\left(\nabla_b\phi\right)\left(\nabla^b\phi\right) - 2\,h^{2a}{}_{a}\left(\nabla_b\phi\right)\left(\nabla^b\phi\right) + \left(h^{1a}{}_{a}\right)^2\left(\nabla_b\phi\right)\left(\nabla^b\phi\right) +$$
$$4\,h^{1}{}_{ba}\left(h^{1ba}\,\mathrm{v}[\phi] + \left(\nabla^a\phi\right)\left(4\,\delta\phi^{1;b} + h^{1c}{}_{c}\left(\nabla^b\phi\right)\right)\right) - 8\,\mathrm{n}\,h^{1b}{}_{b}\,\delta\phi^{1}\,\mathrm{m}'[\phi] - 16\,\delta\mathrm{n}^{1}\,\delta\phi^{1}\,\mathrm{m}'[\phi] -$$
$$8\,\mathrm{n}\,\delta\phi^{2}\,\mathrm{m}'[\phi] - 8\,h^{1b}{}_{b}\,\delta\phi^{1}\,\mathrm{v}'[\phi] - 8\,\delta\phi^{2}\,\mathrm{v}'[\phi] - 8\,\mathrm{n}\left(\delta\phi^{1}\right)^2\mathrm{m}''[\phi] - 8\left(\delta\phi^{1}\right)^2\mathrm{v}''[\phi]\Big)$$

to take the derivative with respect to the scalar field perturbation to get the first order equation of motion

In[22]:=
```
eoms1 = VarD[pertsf[LI[1]], cd][varL2] /. {delta[-LI[1], LI[1]] → 1, delta[-LI[2], LI[1]] → 0,
     delta[-LI[1], LI[2]] → 0} // ToCanonical // Simplification // FullSimplify
```

Out[22]=

$$2\,\delta\phi^{1;a}{}_{;a} + \left(\nabla^a\phi\right)\left(h^{1b}{}_{b;a} - 2\,h^{1}{}_{a}{}^{b}{}_{;b}\right) - 2\,h^{1ab}\left(\nabla_b\nabla_a\phi\right) +$$
$$h^{1a}{}_{a}\left(\nabla_b\nabla^b\phi - \mathrm{nm}'[\phi] - \mathrm{v}'[\phi]\right) - 2\left(\delta\mathrm{n}^{1}\,\mathrm{m}'[\phi] + \left(\delta\phi^{1}\right)\left(\mathrm{nm}''[\phi] + \mathrm{v}''[\phi]\right)\right)$$

that can be written as

$$0 = \left(-\rho''_{DM} - V'' + \Box\right)\delta\phi - m' \cdot \delta n + \frac{1}{2}\left(-\rho'_{DM} - V' + \Box\,\phi + \nabla^{\mu}\phi \cdot \nabla_{\mu}\right) h^{\nu}_{\nu} - h^{\nu}_{\mu;\nu}\,\phi^{;\mu} - h^{\mu\nu}\,\phi_{;\nu\mu},$$

with δn the dark matter number density perturbation, $\phi^{;\mu} = \nabla^{\mu}\phi$, $\delta\phi^{;\mu}_{;\mu} = \Box\,\delta\phi$, and $h_{\mu\nu}$ the metric perturbation.

Stress–Energy Tensor

From the conservation equation of the stress–energy tensor $\nabla_{\mu} T^{\mu\nu} = 0 = \nabla_{\mu}\left(\frac{2}{\sqrt{-g}}\frac{\delta\left(\sqrt{-g}\,L\right)}{\delta g_{\mu\nu}}\right)$ we obtain the stress–energy tensor itself:

In[23]:=
```
DefTensor[myt[-a, -b], M, PrintAs → "T"](* We define the stress-energy tensor T *)
DefTensorPerturbation[pertT[LI[order], -a, -b], myt[-a, -b], M, PrintAs → "δT"]
(* We define the stress-energy perturbations δT *)
DefTensor[myu[-a], M, PrintAs → "u"]
(* The Cosmic Linear Anisotropy Solving System (CLASS) uses the perfect fluid approximation and the
   formalism by Ma & Bertschinger (arXiv:astro-ph/9506072), which includes the four-velocity u *)
DefTensorPerturbation[pertu[LI[order], -a], myu[-a], M, PrintAs → "δu"]
(* The perturbations of the four-velocity *)
```

The stress–energy tensor is $T_{\mu\nu} = \nabla_{\mu}\phi\nabla_{\nu}\phi - \frac{1}{2}\,g_{\mu\nu}\left(\nabla^{\gamma}\phi\nabla_{\gamma}\phi + 2\,V(\phi)\right) - g_{\mu\nu}\,\rho$

In[27]:=
```
set[-a, -b] =
 2 (-VarD[pert[LI[1], a, b], cd][varL] / Sqrt[-Detbackground[]] /. delta[-LI[1], LI[1]] → 1 //
     SeparateMetric[background] // RicciToEinstein) // Expand // ContractMetric // ToCanonical
```

Out[27]=

$$g_{ab}\,m[\phi]\,n + g_{ab}\,V[\phi] - (\nabla_a \phi)(\nabla_b \phi) + \frac{1}{2}\,g_{ab}\,(\nabla_c \phi)(\nabla^c \phi)$$

To make use of the perfect fluid approximation and to be aligned with the choice of CLASS to use the formalism by Ma & Bertschinger we replace $-g_{\mu\nu}$ with $u_{\mu}\,u_{\nu}$ for the DM term and find $T_{\mu\nu} = \nabla_{\mu}\phi\nabla_{\nu}\phi - \frac{1}{2}\,g_{\mu\nu}\left(\nabla^{\gamma}\phi\nabla_{\gamma}\phi + 2\,V(\phi)\right) + \rho\,u_{\mu}\,u_{\nu}$

In[28]:=
```
myt[-a, -b] = cd[-a][sf[]] × cd[-b][sf[]] -
  background[-a, -b] / 2 (cd[c][sf[]] × cd[-c][sf[]] + 2 V[sf[]]) + myu[-a] × myu[-b] × mf[sf[]] × nf[]
```

Out[28]=

$$m[\phi]\,u_a\,u_b\,n + (\nabla_a \phi)(\nabla_b \phi) - \frac{1}{2}\,g_{ab}\left(2\,V[\phi] + (\nabla_c \phi)(\nabla^c \phi)\right)$$

In the perfect fluid approximation using the Ma & Bertschinger notation, the number of dark matter particles is conserved, i.e. $\nabla_{\mu}\left(n\,u^{\mu}\right) = 0$, which we turn in to a rule that can later be applied to simplify equations.

In[29]:=
```
numcons = MakeRule[{cd[a][nf[]] × myu[-a], -nf[] × cd[a][myu[-a]]}]
```

Out[29]=

$$\left\{\text{HoldPattern}\left[\left(\nabla^{\underline{a}} n\right) u_{\underline{a}}\right] :\to \text{Module}\left[\{b\}, -n\,u_b{}^{;b}\right], \text{HoldPattern}\left[\left(\nabla_{\underline{a}} n\right) u^{\underline{a}}\right] :\to \text{Module}\left[\{b\}, -n\,u_b{}^{;b}\right]\right\}$$

The conservation equation of the stress–energy tensor $\nabla_{\mu} T^{\mu\nu} = 0$ yields additional equations of motion,

In[30]:=
```
eomt0 = Simplify[cd[a][myt[-a, -b]], eom0 == 0] /. numcons // Expand // ContractMetric // ToCanonical //
  Simplification
```

Out[30]=

$$m[\phi]\,u^a\,n\,u_{b;a} + \left(\nabla_a \nabla^a \phi\right)(\nabla_b \phi) + u^a\,u_b\,n\,(\nabla_a \phi)\,m'[\phi] - (\nabla_b \phi)\,V'[\phi]$$

respectively $0 = u^{\mu}\nabla_{\mu} u_{\nu} + \frac{m'}{m}\left(\nabla_{\nu}\phi + u_{\nu}\,u_{\mu}\nabla^{\mu}\phi\right)$.

First Order Stress-Energy Tensor

To get the first-order perturbations, we express our stress–energy tensor in terms of the first-order perturbations

In[31]:=
```
myt[-a, -b] // Perturbation // ExpandPerturbation // ContractMetric // ToCanonical // Simplification
```

Out[31]=

$$m[\phi]\left(u_b\, n\, \delta u^1{}_a + u_a\left(u_b\, \delta n^1 + n\, \delta u^1{}_b\right)\right) + (\nabla_a \phi)\, \delta\phi^1{}_{;b} + \delta\phi^1{}_{;a}\,(\nabla_b \phi) - g_{ab}\,(\nabla_c \phi)\, \delta\phi^{1;c} - \frac{1}{2} h^1{}_{ab}\left(2 V[\phi] + (\nabla_c \phi)(\nabla^c \phi)\right) + \frac{1}{2} g_{ab}\, h^1{}_{cd}\,(\nabla^c \phi)(\nabla^d \phi) + u_a\, u_b\, n\, \delta\phi^1\, m'[\phi] - g_{ab}\, \delta\phi^1\, V'[\phi]$$

and use the conservation equation of the stress–energy tensor, the equations of motion to zeroth and first order we gathered so far, as well as the number conservation of dark matter, to find the resulting first-order equations of motion

In[32]:=
```
FullSimplify[
 (Simplify[cd[a][%], {eom0 == 0, eomt0 == 0, eoms1 == 0}] // Expand // ContractMetric // ToCanonical //
    Simplification) /. numcons]
```

Out[32]=

$$-V[\phi]\, h^1{}_b{}^a{}_{;a} + m[\phi]\Big(\left(u^a\, \delta n^1 + n\, \delta u^{1a}\right) u_{b;a} + u_b\left(\delta n^1\, u^a{}_{;a} + \delta u^{1a}\,(\nabla_a n) + u^a\, \delta n^1{}_{;a} + n\, \delta u^{1a}{}_{;a}\right) + u^a\, n\, \delta u^1{}_{b;a} + n\, \delta u^1{}_b\left(u^a{}_{;a} - u_c{}^{;c}\right)\Big) +$$
$$\delta\phi^1{}_{;b}\left(\nabla_a \nabla^a \phi - V'[\phi]\right) + \frac{1}{2}\left(2 n\left(u^a\, \delta\phi^1\, u_{b;a} + u_b\left(u^a\, \delta\phi^1{}_{;a} + \delta\phi^1\left(u^a{}_{;a} - u_d{}^{;d}\right)\right)\right)\right) m'[\phi] +$$
$$(\nabla^a \phi)\left(-2\, h^1{}_b{}^c\,(\nabla_c \nabla_a \phi) + 2\, h^1{}_a{}^c\,(\nabla_c \nabla_b \phi) + h^1{}_{ac;b}\,(\nabla^c \phi) - 2\, h^1{}_{ba}\, V'[\phi]\right) +$$
$$(\nabla_a \phi)\left(-(\nabla^a \phi)\, h^1{}_b{}^c{}_{;c} + 2\left(u_b\left(u^a\, \delta n^1 + n\, \delta u^{1a}\right) + u^a\, n\, \delta u^1{}_b\right) m'[\phi] + 2\, u^a\, u_b\, n\, \delta\phi^1\, m''[\phi]\right)) +$$
$$(\nabla_b \phi)\left(\delta\phi^{1;a}{}_{;a} - \delta\phi^1\, V''[\phi]\right)$$

Einstein Source Equation

The temporal and spatial parts of the stress–energy tensor yield the Einstein source equations. This requires us to choose a metric, which we will do in the next steps in the `xPand.nb`. However, with some hindsight we can find them in the components of T already.

Dark Matter Density (Perturbation)

The time–time component of the stress–energy tensor T^0_0 gives us $\rho_{DM} = m\, n$

In[33]:=
```
rhocdm = mf[sf[]] × nf[]
```

Out[33]=

$$m[\phi]\, n$$

respectively the density perturbation $\delta\rho = \rho\left(\frac{\delta n}{n} + \frac{m'}{m}\,\delta\phi\right)$

In[34]:=
```
drhocdm = rhocdm // Perturbation // ExpandPerturbation // ContractMetric // ToCanonical
```

Out[34]=

$$m[\phi]\, \delta n^1 + n\, \delta\phi^1\, m'[\phi]$$

Scalar Field Density (Perturbation)

Furthermore, the temporal part of the stress–energy tensor T^0_0 contains additional information, namely the scalar field density $\rho_\phi = \nabla_0\phi\nabla_0\phi - \frac{1}{2}\, g_{00}\left(\nabla^0\phi\nabla_0\phi + 2\, V\right) = \frac{(\partial_\eta \phi)^2}{2\, a^2} + V$, where we used that the scalar field is spatially homogenous and depends only on time.

In[35]:=
```
rhophi = (V[sf[]] + 1/2 cd[-b][sf[]] × cd[b][sf[]])
```
Out[35]=

$$V[\phi]+\frac{1}{2}(\nabla_b\phi)(\nabla^b\phi)$$

The perturbations yields $\delta\rho_\phi = \frac{1}{a^2}\left(\partial_\eta\phi\cdot\partial_\eta\delta\phi - \alpha\,(\partial_\eta\phi)^2\right) + V'\cdot\delta\phi$

In[36]:=
```
drhophi = (rhophi) // Perturbation // ExpandPerturbation // ContractMetric // ToCanonical
```
Out[36]=

$$(\nabla_b\phi)\,\delta\phi^{1;b}-\frac{1}{2}\,h^1{}_{ba}\,(\nabla^a\phi)(\nabla^b\phi)+\delta\phi^1\,V'[\phi]$$

Scalar Field Pressure (Perturbation)

From the spatial part we can read of the scalar field density $p_\phi = \frac{(\partial_\eta\phi)^2}{2\,a^2} - V$

In[37]:=
```
pphi = (-V[sf[]] + 1/2 cd[-b][sf[]] × cd[b][sf[]])
```
Out[37]=

$$-V[\phi]+\frac{1}{2}(\nabla_b\phi)(\nabla^b\phi)$$

which we can perturb to obtain the scalar field pressure perturbation $\delta\,p_\phi = \frac{1}{a^2}\left(\partial_\eta\phi\cdot\partial_\eta\delta\phi - \alpha\,(\partial_\eta\phi)^2\right) - V'\cdot\delta\phi$

In[38]:=
```
dpphi = (pphi) // Perturbation // ExpandPerturbation // ContractMetric // ToCanonical
```
Out[38]=

$$(\nabla_b\phi)\,\delta\phi^{1;b}-\frac{1}{2}\,h^1{}_{ba}\,(\nabla^a\phi)(\nabla^b\phi)-\delta\phi^1\,V'[\phi]$$

Energy Flux

From the time–space components of the stress–energy tensor, we obtain the energy flux $T_i^0 = -T_0^i = (\bar{\rho}+\bar{p})\,u_i$

In[39]:=
```
rhoplusp = (rhophi + pphi) myu[-a]
```
Out[39]=

$$u_a\,(\nabla_b\phi)(\nabla^b\phi)$$

If we take the spatial derivative of the spatial components of this equation, we obtain the scalar field velocity divergence. In CLASS, this term is called `rho_plus_p_theta`, with $\partial^i u_i = \theta$, following the notation of Ma & Bertschinger. As we can see, this term will not depend on the scalar field potential nor on the dark matter mass. Therefore, it won't require a change in the CLASS source code and we can keep the `rho_plus_p_theta`-term untouched.

We have now a complete set of equations that need to be expressed depending on a metric as well as a gauge, to be implemented into CLASS. We obtain those equations in the `Mathematica` notebook `xPand.nb`.