Attractor Solutions / Tracking Conditions

Quintessence has early and late time tracking / attractor solutions, where the scalar field ‘mimics’ the background energy density (early time behaviour) respectively approaches a plateau (late time). For an attractor solution’s equation of state, $\omega_\phi = \omega_{\text{background}}$ holds. For a scalar field we have $\omega_\phi = \frac{p_\phi}{\rho_\phi} = \frac{\frac{1}{2}\dot{\phi}^2 - V}{\frac{1}{2}\dot{\phi}^2 + V}$.

In[ ]:=

```
phiDotSquared = First[Solve[ω == (x - 2 V) / (x + 2 V), x]]
```

Out[ ]=

$$\left\{x \to -\frac{2\,(V + V\,\omega)}{-1 + \omega}\right\}$$

For an attractor solution, $\dot{\phi}^2 = 2\,V\,\frac{1+\omega}{1-\omega}$ holds. Therefore, $\rho_\phi = \frac{2\,V}{1-\omega}$:

In[ ]:=

```
ρ = (x / 2 + V) /. phiDotSquared // Simplify
```

Out[ ]=

$$-\frac{2\,V}{-1 + \omega}$$

To express this in terms of the potential V and the scalar field ϕ, we need the Friedmann equation $3\,H^2 = \sum_i \rho_i$, the definition for the relative energy density $\Omega_\phi = \frac{\rho_\phi}{\rho_{\text{crit}}} = \frac{\rho_\phi}{3\,H^2}$, and the result from Copeland et al. 1998 (arXiv:gr-qg/9711068) $\Omega_\phi = \frac{3\,\gamma}{(V'/V)^2}$, with $\gamma = 4/3$ for radiation, $\gamma = 1$ for dust, and $\gamma = 1 + \omega$ in general.

In[ ]:=

```
Solve[Eliminate[{3 H^2 == p + b, Ω == p / (3 H^2), Ω == 3 γ / λ^2}, {H, Ω}], p]
```

Out[ ]=

$$\left\{\left\{p \to -\frac{3\, b\, \gamma}{3\, \gamma - \lambda^2}\right\}\right\}$$

We find $\rho_\phi = \rho_{\text{background}} \dfrac{3\,\gamma/\lambda^2}{1-3\,\gamma/\lambda^2}$. To find our initial conditions for ϕ and ϕ', we have to work with explicit potentials, which we do in the following.

Setup

Potentials

In[ ]:=

```
potentials = <|
    "pl" → c^(4 - p) ϕ^p,
    "cos" → c1 Cos[c2 ϕ],
    "hyp" → c1 (1 - Tanh[c2 ϕ]),
    "pNG" → c1^4 (1 + Cos[ϕ / c2]),
    "iPL" → c^(p + 4) ϕ^(-p),
    "exp" → c1 Exp[-c2 ϕ],
    "sqexp" → c^(p + 4) ϕ^(-p) Exp[c ϕ^2],
    "bexp" → c1 (ϕ^2 + c2) Exp[-c3 ϕ],
    "dexp" → c1 Exp[-c2 ϕ] + c3 Exp[-c4 ϕ] |>;

vars = {ϕ, p, c, c1, c2, c3, c4};
names = Keys[potentials];
vals = Values[potentials];
```

Initial Setup for Parallel Computations

In[ ]:=
```
LaunchKernels[];
DistributeDefinitions[potentials, ω, ρ, γ, c, p, c1, c2, c3, c4];
```

Exact Computations

Existence of Attractor Solution

An attractor solution exists if $\Gamma = \frac{V\,V''}{(V')^2} \geq 1$.

In[ ]:=

```
exactGamma = ParallelTable[With[{V = potentials[[name]]}, V D[V, {ϕ, 2}] / (D[V, ϕ])^2], {name, Keys[potentials]}];
gammalimitSmallPhi = ParallelTable[Limit[exactGamma[[i]], ϕ → 0], {i, Length[exactGamma]}];
gammalimitLargePhi = ParallelTable[Limit[exactGamma[[i]], ϕ → Infinity], {i, Length[exactGamma]}];
TableForm[Transpose[{Keys[potentials], exactGamma, gammalimitSmallPhi, gammalimitLargePhi}],
  TableHeadings → {None, {"Potential", "Γ=VV''/(V')^2", "ϕ->0 Γ", "ϕ->∞ Γ"}}]
```

Out[ ]//TableForm=

Potential	Γ=VV''/(V')^2	ϕ->0 Γ	ϕ->∞ Γ
pl	$\frac{-1+p}{p}$	$\frac{-1+p}{p}$	$\frac{-1+p}{p}$
cos	$-\mathrm{Cot}[c2\,\phi]^2$	$\mathrm{DirectedInfinity}\left[-\frac{1}{\mathrm{Sign}[c2]^2}\right]$	Indeterminate if c2 > 0
hyp	$2\,\mathrm{Cosh}[c2\,\phi]\,\mathrm{Sinh}[c2\,\phi]\,(1-\mathrm{Tanh}[c2\,\phi])$	0	1 if c2 > 0
pNG	$-\left(\left(1+\mathrm{Cos}\left[\frac{\phi}{c2}\right]\right)\mathrm{Cot}\left[\frac{\phi}{c2}\right]\mathrm{Csc}\left[\frac{\phi}{c2}\right]\right)$	$\mathrm{DirectedInfinity}[-\mathrm{Sign}[c2]^2]$	Indeterminate if c2 > 0
iPL	$-\frac{-1-p}{p}$	$-\frac{-1-p}{p}$	$-\frac{-1-p}{p}$
exp	1	1	1
sqexp	$\frac{c^{4+p}\,e^{c\phi^2}\,\phi^{-p}\left(-c^{4+p}\,e^{c\phi^2}\,(-1-p)\,p\,\phi^{-2-p}-4\,c^{5+p}\,e^{c\phi^2}\,p\,\phi^{-p}+c^{4+p}\,\phi^{-p}\left(2\,c\,e^{c\phi^2}+4\,c^2\,e^{c\phi^2}\,\phi^2\right)\right)}{\left(-c^{4+p}\,e^{c\phi^2}\,p\,\phi^{-1-p}+2\,c^{5+p}\,e^{c\phi^2}\,\phi^{1-p}\right)^2}$	$1+\frac{1}{p}$	1
bexp	$\frac{c1\,e^{-c3\,\phi}\,(c2+\phi^2)\,(2\,c1\,e^{-c3\,\phi}-4\,c1\,c3\,e^{-c3\,\phi}\,\phi+c1\,c3^2\,e^{-c3\,\phi}\,(c2+\phi^2))}{(2\,c1\,e^{-c3\,\phi}\,\phi-c1\,c3\,e^{-c3\,\phi}\,(c2+\phi^2))^2}$	$\frac{2\,c1+c1\,c2\,c3^2}{c1\,c2\,c3^2}$	1
dexp	$\frac{(c1\,e^{-c2\,\phi}+c3\,e^{-c4\,\phi})\,(c1\,c2^2\,e^{-c2\,\phi}+c3\,c4^2\,e^{-c4\,\phi})}{(-c1\,c2\,e^{-c2\,\phi}-c3\,c4\,e^{-c4\,\phi})^2}$	$\frac{(c1+c3)\,(c1\,c2^2+c3\,c4^2)}{(-c1\,c2-c3\,c4)^2}$	1 if $(c1 \mid c3) \in \mathbb{R}$ && c2 > 0 &

We see that power-law potentials, Cosines, and pseudo-Nambu--Goldstone potentials do not have attractor solutions. These are thawing respectively oscillating scenarios. In CLASS, we will have to brute-force the initial conditions for ϕ and ϕ' to match the energy budget. All other potentials have attractor solutions for large ϕ, and except for the hyperbolic potential also for small ϕ.

An important quantity is the relative slope of the potential, $\lambda = \frac{V'}{V}$:

In[ ]:=
```
exactlambda = ParallelTable[With[{V = potentials[[name]]}, D[V, ϕ] / V], {name, Keys[potentials]}];
```

as well as the tracking density $\rho_{\text{tracking}} = \rho_{\text{background}} \frac{3\,\gamma}{\lambda - 3\,\gamma}$

In[ ]:=
```
rhoTracking = ρ 3 γ / exactlambda / (1 - 3 γ / exactlambda);
```

where we use ρ as the background energy density and γ=1 for dust or γ=4/3 for radiation, i.e. γ=1+ω.
This then yields the initial $\phi' = \sqrt{\rho_{\text{tracking}}(1+\omega)}$:

In[ ]:=
```
phiPrimeIni = Sqrt[rhoTracking (1 + ω)] // FullSimplify;
```

In[ ]:=

```
TableForm[Transpose[{Keys[potentials], exactlambda, phiPrimeIni}],
  TableHeadings → {None, {"Potential", "λ = V' / V", "initial ϕ'"}}]
```

Out[ ]//TableForm=

Potential	λ = V' / V	initial ϕ'
pl	$\frac{p}{\phi}$	$\sqrt{3}\ \sqrt{\frac{\gamma\ \rho\ \phi\ (1+\omega)}{p-3\ \gamma\ \phi}}$
cos	$-c2\ \text{Tan}[c2\ \phi]$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)\ \text{Cot}[c2\ \phi]}{c2+3\ \gamma\ \text{Cot}[c2\ \phi]}}$
hyp	$-\frac{c2\ \text{Sech}[c2\ \phi]^2}{1-\text{Tanh}[c2\ \phi]}$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)\ \text{Cosh}[c2\ \phi]}{(c2+3\ \gamma)\ \text{Cosh}[c2\ \phi]+c2\ \text{Sinh}[c2\ \phi]}}$
pNG	$-\frac{\text{Sin}\left[\frac{\phi}{c2}\right]}{c2\left(1+\text{Cos}\left[\frac{\phi}{c2}\right]\right)}$	$\sqrt{\rho\ (1+\omega)\left(-1+\frac{1}{1+3\ c2\ \gamma\ \text{Cot}\left[\frac{\phi}{2\ c2}\right]}\right)}$
iPL	$-\frac{p}{\phi}$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ \phi\ (1+\omega)}{p+3\ \gamma\ \phi}}$
exp	$-c2$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{c2+3\ \gamma}}$
sqexp	$c^{-4-p}\ e^{-c\ \phi^2}\ \phi^p\left(-c^{4+p}\ e^{c\ \phi^2}\ p\ \phi^{-1-p}+2\ c^{5+p}\ e^{c\ \phi^2}\ \phi^{1-p}\right)$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ \phi\ (1+\omega)}{p+\phi\ (3\ \gamma-2\ c\ \phi)}}$
bexp	$\frac{e^{c3\ \phi}\left(2\ c1\ e^{-c3\ \phi}\ \phi-c1\ c3\ e^{-c3\ \phi}\ (c2+\phi^2)\right)}{c1\ (c2+\phi^2)}$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (c2+\phi^2)\ (1+\omega)}{-2\ \phi+(c3+3\ \gamma)\ (c2+\phi^2)}}$
dexp	$\frac{-c1\ c2\ e^{-c2\ \phi}-c3\ c4\ e^{-c4\ \phi}}{c1\ e^{-c2\ \phi}+c3\ e^{-c4\ \phi}}$	$\sqrt{3}\ \sqrt{-\frac{(c3\ e^{c2\ \phi}+c1\ e^{c4\ \phi})\ \gamma\ \rho\ (1+\omega)}{c1\ e^{c4\ \phi}\ (c2+3\ \gamma)+c3\ e^{c2\ \phi}\ (c4+3\ \gamma)}}$

We see that most of these attractor initial conditions depend on the initial field value. We can take large field attractors, which are in a quintessence context usually understood as the early universe attractor, and a small-field, late-time attractor solution.

Large ϕ Attractor Solutions

Initial ϕ'

In the large ϕ limit, we obtain the following initial values for ϕ':

In[ ]:=

```
phiPrimeIniLargePhi = ParallelTable[Limit[phiPrimeIni[[i]], ϕ → Infinity] // Simplify, {i, Length[phiPrimeIni]}];
```

In[ ]:=

```
lambdaLimitLargePhi = ParallelTable[Limit[exactlambda[[i]], ϕ → Infinity], {i, Length[exactlambda]}];
```

In[ ]:=

```
TableForm[Transpose[{Keys[potentials], lambdaLimitLargePhi, phiPrimeIniLargePhi}],
  TableHeadings → {None, {"Potential", "large−field λ", "large−field ϕ′ini"}}]
```

Out[ ]//TableForm=

Potential	large−field λ	large−field ϕ'_{ini}
pl	0	$\sqrt{-\rho\ (1+\omega)}$
cos	Indeterminate if $c2 \in \mathbb{R}$	$\lim_{\phi\to\infty} \sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)\ \mathrm{Cot}[c2\ \phi]}{c2+3\ \gamma\ \mathrm{Cot}[c2\ \phi]}}$
hyp	$-2\ c2$ if $c2>0$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{2\ c2+3\ \gamma}}$ if $(\gamma \mid \rho \mid \omega) \in \mathbb{R}\ \&\&\ c2>0\ \&\&\ \gamma\ \rho\ (1+\omega)\ \mathrm{Re}\left[\frac{1}{2\ c2+3\ \gamma}\right]<0$
pNG	$\lim_{\phi\to\infty} -\frac{\mathrm{Sin}\left[\frac{\phi}{c2}\right]}{c2\left(1+\mathrm{Cos}\left[\frac{\phi}{c2}\right]\right)}$	$\lim_{\phi\to\infty} \sqrt{\rho\ (1+\omega)\left(-1+\frac{1}{1+3\ c2\ \gamma\ \mathrm{Cot}\left[\frac{\phi}{2\ c2}\right]}\right)}$
iPL	0	$\sqrt{-\rho\ (1+\omega)}$
exp	$-c2$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{c2+3\ \gamma}}$
sqexp	$c\ \infty$	0
bexp	$-c3$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{c3+3\ \gamma}}$
dexp	$-c4$ if $(c1 \mid c3) \in \mathbb{R}\ \&\&\ c2>0\ \&\&\ c2>c4\ \&\&\ c4>0$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{c4+3\ \gamma}}$ if condition

We already know that power-law potentials, Cosines, and pseudo-Nambu--Goldstone potentials do not have attractor solutions. Here, we see that Hyperbolic potentials and double exponential potentials do not have physical attractor solutions, as there is no real initial ϕ' for those. However, both approximate a simple exponential, i.e. at least numerically an attractor initial condition exists, which suffices for CLASS.

Initial ϕ

For the initial value of ϕ we solve the equation $\dfrac{2\,V(\phi)}{1-\omega} = \rho_{\text{tracking}}$. This does not always have a real or close-form solution, since ρ_{tracking} depends on V and V' and therefore non-trivially on ϕ itself.

In[ ]:=

```
(*Compute large-phi initial conditions in parallel*)
largePhiInitial = ParallelTable[Module[{name = names[[i]], V = vals[[i]], LambdaLargePhi = lambdaLimitLargePhi[[i]], VLargePhi, eqn,
      result}, (*Asymptotic form of V as ϕ→∞ (leading term only)*) VLargePhi = Normal@Series[V, {ϕ, Infinity, 1}];
     (*Construct attractor equation using asymptotic Lambda and V*) eqn = 2 VLargePhi / (1 - ω) ==
        ρ 3 γ / (LambdaLargePhi)^2 / (1 - 3 γ / (LambdaLargePhi)^2) && p > 0 && γ > 0 && ρ > 0 && ω > 0;
     (*Try to solve for ϕ*) If[name === "iPL",
       (*Use Reduce for iPL*) result = Quiet@Reduce[eqn, ϕ, Reals],
       (*Use Solve for all other potentials*) Module[{soln}, soln = Quiet@Solve[eqn, ϕ, Reals];
        result = If[soln =!= {}, First[soln], "no real solution / ϕ unconstrained"]]];
     {name, result}], {i, Length[vals]}];

(*Display*)
TableForm[Transpose[{names, largePhiInitial[[All, 2]] (*extract solutions*)}],
  TableHeadings → {None, {"Potential", "ϕini (large ϕ attractor)"}}]
```

kernel 1 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 1 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 5 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 1 (Local)

Infinity : Indeterminate expression 0 ComplexInfinity encountered.

kernel 5 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 5 (Local)

Infinity : Indeterminate expression 0 ComplexInfinity encountered.

Out[∘]//TableForm=

Potential	ϕ_{ini} (large ϕ attractor)
pl	no real solution / ϕ unconstrained
cos	no real solution / ϕ unconstrained
hyp	$\phi \to \frac{\text{ArcTanh}\left[\frac{8\,c1\,c2^2 - 6\,c1\,\gamma - 3\,\gamma\,\rho + 3\,\gamma\,\rho\,\omega}{2\,c1\,(4\,c2^2 - 3\,\gamma)}\right]}{c2}$ if condition
pNG	$\frac{2\left(c1^4 + c1^4\,\text{Cos}\left[\frac{\phi}{c2}\right]\right)}{1-\omega} == \frac{3\,\gamma\,\rho}{\left(1 - \frac{3\,\gamma}{\left(\lim_{\phi\to\infty} - \frac{\text{Sin}\left[\frac{\phi}{c2}\right]}{c2\left(1+\text{Cos}\left[\frac{\phi}{c2}\right]\right)}\right)^2}\right)\left(\lim_{\phi\to\infty} - \frac{\text{Sin}\left[\frac{\phi}{c2}\right]}{c2\left(1+\text{Cos}\left[\frac{\phi}{c2}\right]\right)}\right)^2}$ && $p > 0$ && $\gamma > 0$ && $\rho > 0$ && $\omega > 0$
iPL	False
exp	$\phi \to \frac{\text{Log}\left[-\frac{2\,(c1\,c2^2 - 3\,c1\,\gamma)}{3\,\gamma\,\rho\,(-1+\omega)}\right]}{c2}$ if condition
sqexp	no real solution / ϕ unconstrained
bexp	$\frac{2\,e^{-c3\,\phi}\,(c1\,c2 + c1\,\phi^2)}{1-\omega} == \frac{3\,\gamma\,\rho}{c3^2\left(1 - \frac{3\,\gamma}{c3^2}\right)}$ && $p > 0$ && $\gamma > 0$ && $\rho > 0$ && $\omega > 0$
dexp	$\frac{2\,(c1\,e^{-c2\,\phi} + c3\,e^{-c4\,\phi})}{1-\omega} == \frac{3\,\gamma\,\rho}{c4^2\left(1 - \frac{3\,\gamma}{c4^2}\right)}$ && $p > 0$ && $\gamma > 0$ && $\rho > 0$ && $\omega > 0$ if $(c1 \mid c3) \in \mathbb{R}$ && $c2 > 0$ && $c2 > c4$ && $c4 > 0$

We notice that the double exponential potential and Bean’s exponential cannot be solved with the means of Solve (or Reduce), yet both can be approximated by a simple exponential.

Small ϕ Attractors

Here, we compute the late-time, small-field attractors.

Initial ϕ'

In[ ]:=
```
phiPrimeIniSmallPhi = ParallelTable[Limit[phiPrimeIni[[i]], ϕ → 0], {i, Length[phiPrimeIni]}];
```

In[ ]:=
```
lambdaLimitSmallPhi = ParallelTable[Limit[exactlambda[[i]], ϕ → 0], {i, Length[exactlambda]}];
```

In[ ]:=

```
TableForm[Transpose[{Keys[potentials], lambdaLimitSmallPhi, phiPrimeIniSmallPhi}],
  TableHeadings → {None, {"Potential", "small−field λ", "small−field ϕ'ini"}}]
```

Out[ ]//TableForm=

Potential	small−field λ	small−field ϕ'_{ini}
pl	Indeterminate	0
cos	0	$\sqrt{-\rho\ (1+\omega)}$
hyp	$-\text{c2}$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{\text{c2}+3\ \gamma}}$
pNG	0	$\sqrt{-\rho\ (1+\omega)}$
iPL	Indeterminate	0
exp	$-\text{c2}$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{\text{c2}+3\ \gamma}}$
sqexp	Indeterminate	0
bexp	$-\text{c3}$	$\sqrt{3}\ \sqrt{-\frac{\gamma\ \rho\ (1+\omega)}{\text{c3}+3\ \gamma}}$
dexp	$\frac{-\text{c1}\ \text{c2}-\text{c3}\ \text{c4}}{\text{c1}+\text{c3}}$	$\sqrt{3}\ \sqrt{-\frac{(\text{c1}+\text{c3})\ \gamma\ \rho\ (1+\omega)}{\text{c1}\ (\text{c2}+3\ \gamma)+\text{c3}\ (\text{c4}+3\ \gamma)}}$

We see that Bean’s exponential and the double exponential have only conditional solutions. If the conditions are not met, we can still approximate those as simple exponential potentials.

Initial ϕ

In[ ]:=

```
(*Compute small-phi initial conditions in parallel*)
smallPhiInitial = ParallelTable[Module[{name = names[[i]], V = vals[[i]], LambdaSmallPhi = lambdaLimitSmallPhi[[i]], VSmallPhi, eqn,
      result}, (*Asymptotic form of V as ϕ→0 (leading term only) *) VSmallPhi = Normal@Series[V, {ϕ, 0, 1}];
    (*Construct attractor equation using asymptotic Lambda and V*) eqn = 2 VSmallPhi / (1 - ω) ==
        ρ 3 γ / (LambdaSmallPhi)^2 / (1 - 3 γ / (LambdaSmallPhi)^2) && p > 0 && γ > 0 && ρ > 0 && ω > 0;
    (*Try to solve for ϕ*) If[name === "iPL", (*Use Reduce for iPL*) result = Quiet@Reduce[eqn, ϕ, Reals],
      (*Use Solve for all other potentials*) Module[{soln}, soln = Quiet@Solve[eqn, ϕ, Reals];
       result = If[soln =!= {}, First[soln], "no real solution / ϕ unconstrained"]]];
    {name, result}], {i, Length[vals]}];

(*Display*)
TableForm[Transpose[{names, smallPhiInitial[[All, 2]] (*extract solutions*)}],
  TableHeadings → {None, {"Potential", "small ϕ_ini attractor"}}]
```

kernel 2 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 4 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 2 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 4 (Local)

Power : Infinite expression $\frac{1}{0}$ encountered.

kernel 2 (Local)

Infinity : Indeterminate expression 0 ComplexInfinity encountered.

kernel 4 (Local)

Infinity : Indeterminate expression 0 ComplexInfinity encountered.

Out[ ]//TableForm=

Potential	small ϕ_{ini} attractor
pl	no real solution / ϕ unconstrained
cos	no real solution / ϕ unconstrained
hyp	$\phi \to \frac{2\,c1\,c2^2 - 6\,c1\,\gamma - 3\,\gamma\,\rho + 3\,\gamma\,\rho\,\omega}{2\,c1\,c2^3 - 6\,c1\,c2\,\gamma}$ if $p > 0\ \&\&\ \gamma > 0\ \&\&\ \omega > 0\ \&\&\ \rho > 0$
pNG	no real solution / ϕ unconstrained
iPL	False
exp	$\phi \to \frac{2\,c1\,c2^2 - 6\,c1\,\gamma - 3\,\gamma\,\rho + 3\,\gamma\,\rho\,\omega}{2\,c1\,c2^3 - 6\,c1\,c2\,\gamma}$ if $p > 0\ \&\&\ \gamma > 0\ \&\&\ \omega > 0\ \&\&\ \rho > 0$
sqexp	no real solution / ϕ unconstrained
bexp	$\phi \to \frac{2\,c1\,c2\,c3^2 - 6\,c1\,c2\,\gamma - 3\,\gamma\,\rho + 3\,\gamma\,\rho\,\omega}{2\,c1\,c2\,c3^3 - 6\,c1\,c2\,c3\,\gamma}$ if $p > 0\ \&\&\ \gamma > 0\ \&\&\ \omega > 0\ \&\&\ \rho > 0$
dexp	$\phi \to (2\,c1^3\,c2^2 + 2\,c1^2\,c2^2\,c3 + 4\,c1^2\,c2\,c3\,c4 + 4\,c1\,c2\,c3^2\,c4 + 2\,c1\,c3^2\,c4^2 + 2\,c3^3\,c4^2 - 6\,c1^3\,\gamma - 18\,c1^2\,c3\,\gamma - 18\,c1\,c3^2\,\gamma - 6\,c3^3\,\gamma - 3\,c1^2\,\gamma\,\rho - 6\,c1\,c3\,\gamma\,\rho - 3\,c3^2\,\gamma\,\rho + 3\,c1^2\,\gamma\,\rho\,\omega + 6\,c1\,c3\,\gamma\,\rho\,\omega + 3\,c3^2\,\gamma\,\rho\,\omega) / (2\,c1^3\,c2^3 + 6\,c1^2\,c2^2\,c3\,c4 + 6\,c1\,c2\,c3^2\,c4^2 + 2\,c3^3\,c4^3 - 6\,c1^3\,c2\,\gamma - 12\,c1^2\,c2\,c3\,\gamma - 6\,c1\,c2\,c3^2\,\gamma - 6\,c1^2\,c3\,c4\,\gamma - 12\,c1\,c3^2\,c4\,\gamma - 6\,c3^3\,c4\,\gamma)$ if $p > 0\ \&\&\ \gamma > 0\ \&\&\ \omega > 0\ \&\&\ \rho > 0$

The conditions are always satisfied in our cosmological setting. This means that either there is no solution for a given potential or that there is always a solution.